
Project 1: Instructions (Part 3)
Learning Code

1
IDE

An IDE, or integrated development 

enviornment, is a software that lets you 

write and run code within one place

2
Header Files

These files act as a library that the 

computer reads so it knows how to 

perform a certain function

#include <Arduino.h>

#include <Wire.h>

#include <Adafruit_GFX.h>

3
Setup Function

This function is where your computer 

reads the libraries you provide and gives 

initial values before running any actions

void setup() {

  pinMode(trigPin, OUTPUT); pinMode(echoPin, INPUT);

  pinMode(ledPin, OUTPUT); pinMode(buzzerPin, OUTPUT);

  Serial.begin(9600);analogReadResolution(10);

}

4
Loop Function

This function is where your computer runs 

the code and tells the connected 

electronics what to do

void loop() {

  lightValue = analogRead(photoPin);

  distance = getDistance();

  Serial.print("Light: "); Serial.print(lightValue);

  Serial.print("   Distance: "); Serial.println(distance);

5
Other Functions

Here are some other common functions 

worth exploring:

While

If-Else

Switch

Custom Functions

Testing the Project

1 OLED Display should turn ON

2 Press button to start game

3 Use joystick to move character on 

screen left and right

4 Buzzer and Servo should activate, 

eventually stopping

5 Use button to “shift gears”, changing 

buzzer and servo position

Troubleshooting

Mini OLED not on

Button not responding

Servo not moving

Success! You've completed your first 

embedded systems project. Try modifying 

the code to add LED blinking or multiple 

button inputs.

Challenge Ideas

• Add the second Joystick direction toggles +/- Y

• Make an LED blink to indicate a gear shift

• Add an acceleration physics feature like a real car

06 Mini Project 2

Advance your skills with a multi-component project: a passive nightlight that uses motion 

detection and a photoresistor so it turns on when you need it.

Project 2: Night Light

Project Description

Build a system that uses motion detection and the light-

sensing photo resistor to turn on an LED nightlight and a 

small chime. Using a while loop and an if-else statement, 

it is an easy project to learn the basics of software code.

Learning Objectives

• Reading analog sensor data (ultrasonic)

• Conditional logic and thresholds

• Controlling multiple outputs simultaneously

• Generating audio alerts

Difficulty: Beginner Time: 30 minutes

Components Needed

Photoresistor

Ultrasonic sensor (HC-SR04)

Buzzer/Speaker

LED (red)

Resistors (see below)

Jumper wires

Breadboard

Battery Pack

Transistor

Resistor Values: 

Blue: 10k Ω

Pink: 470 Ω

Green: 1.8k Ω

Red: 3.3k Ω

Orange: 1k Ω

System Diagram:

Project 2: Instructions (Part 1)

Hardware Assembly & Setup

1
Connect Ultrasonic Sensor

Connect VCC to 5V, GND to ground, Trig to 

pin 7, Echo to resistor then to pin 9. Also 

add the second resistor from pin 9 to GND

2
Connect Buzzer

Insert the transistor with the flat side 

facing you. Connect the left pin to the 

ground rail. Connect Pin 6 to the 

transistor’s middle pin through a 1 kΩ 

resistor. Then connect the buzzer’s 

negative lead to the transistor’s right pin 

and the buzzer’s positive lead to the lower 

power rail.

3
Photoresistor Setup

Connect photoresistor to A0 (pin 38) 

and then other end to the resistor 

before connecting that to GND

4
Add LED

Connect pins 24, 25, and 26 to the Red, 

Green, and Blue LED legs, respectively, using 

a 470 Ω resistor for each connection. 

Connect the LED’s long leg to the GND rail.

5
Connect Battery

Place battery pack in bottom power 

rails, and put in batteries

6
Common Ground

Use a jumper wire to connect the Teensy 

GND pin to the ground rail. Then, 

connect the two ground rails with 

another jumper wire.

7
Upload Code

Upload the code in the following steps!

Component Checklist

• Battery Pack • Ultrasonic sensor

• Buzzer

• LED

• Photoresistor

• Resistors

• Jumper wires • Breadboard

• Transistor

Pin Summary

Pin 24: LED R

Pin 25: LED G

Pin 26: LED B

Pin 6: Buzzer

Pin 7: Ultrasonic Trig

Pin 9: Ultrasonic Echo

Pin 38 (A0): Photoresistor

Note: Ultrasonic sensor echo outputs 5V. Ensure
voltage divider (2 resistors) is in circuit before pin 9 
connection.

Project 2: Instructions (Part 2)

Programming Steps

7
Create New Sketch

Open Arduino IDE and create a new 

sketch. Name it "Night Light"

8
Define Pins and Constants

int photoPin = A0; int trigPin = 7; int echoPin = 9; int redPin 

= 24; int greenPin = 25; int bluePin = 26; int buzzerPin = 6; 

int lightValue = 0; int darkLevel = 500; int closeDistance = 

50; float distance = 0;

9
Set Up Initial State

void setup() {�

pinMode(trigPin, OUTPUT); pinMode(echoPin, INPUT); 

pinMode(redPin, OUTPUT);�

pinMode(greenPin, OUTPUT); pinMode(bluePin, OUTPUT); 

pinMode(buzzerPin, OUTPUT);�

// Start RGB LED off, setLED(false); Serial.begin(9600); 

analogReadResolution(10);�

}

10
Begin Sensing

void loop() { lightValue = analogRead(photoPin); 

distance = getDistance(); Serial.print("Light: "); 

Serial.print(lightValue); Serial.print(" Distance: "); 

Serial.println(distance);�

if (�

lightValue < darkLevel &&�

distance > 0 &&�

distance < closeDistance) {�

// Same behavior as old single-color LED: turn the light 

ON. All three RGB channels create white.�

setLED(true);�

if (!alreadyTriggered) { playLullaby(); alreadyTriggered = 

true;�

}}

11
If-Else Loop

Checks the variable is true or false. If it is a 

true statement, the if code runs. False 

statement, then the else code runs.

if (state) {�

digitalwrite(redPin, HIGH);�

digitalWrite(greenPin, HIGH);�

digitalWrite(bluePin, HIGH); } �

else {�

digitalWrite(redPin, LOW);�

digitalWrite(greenPin, LOW);�

digitalWrite(bluePin, LOW);}

12
Custom Function

You can also define your own function:

float getDistance() {�

long time;�

digitalWrite(trigPin, LOW);�
delayMicroseconds(2);�

digitalWrite(trigPin, HIGH);�
delayMicroseconds(10);�

digitalWrite(trigPin, LOW);�

time = pulseIn(�
echoPin,HIGH,30000);�

return time * 0.034 / 2;�
}

Code Structure

setup(): Initialize pins and code

loop(): Measure distance, check light threshold

If-Else: Checks conditions to decide action

Action: Activate buzzer and LED when triggered

Distance Calculation

Ultrasonic sensors measure time for sound 
to bounce back:

duration = pulseIn(echoPin, HIGH);
distance = duration * 0.034 / 2;

Speed of sound: 343 m/s (0.034 cm/μs)

Used Functions

pinMode() - set pin input/output

Serial.print() - print to terminal

delay() - pause program

getDistance() - ultrasonic sensor

Alarm Logic

distance < threshold: buzzer ON + LED HIGH.
Otherwise: both OFF.

Project 1: Complete Code

Complete Code

#include <Arduino.h>
#include <Wire.h>
#include <Adafruit_GFX.h>
#include <Adafruit_SSD1306.h>
#include <Servo.h>
// 
============================================================
// V1 Racing Game Firmware for Teensy 4.1
// Revised for transistor-driven speaker on separate 5V rail
// ------------------------------------------------------------
// Pins:
// OLED SDA -> 18
// OLED SCL -> 19
// Joystick X-axis -> 14
// Shift button -> 33
// Servo PWM -> 3
// Speaker control -> 10 (to transistor base through 1k)
// 
============================================================
// -------------------- Display --------------------
constexpr int SCREEN_WIDTH = 128;
constexpr int SCREEN_HEIGHT = 64;
constexpr int OLED_RESET = -1;
Adafruit_SSD1306 display(SCREEN_WIDTH, SCREEN_HEIGHT, &Wire, OLED_RESET);
// -------------------- Pins -----------------------
constexpr int JOYSTICK_PIN = 14;
constexpr int BUTTON_PIN = 33;
constexpr int SERVO_PIN = 3;
constexpr int SPEAKER_PIN = 1; // changed from pin 1
// -------------------- Servo ----------------------
Servo rpmServo;
// -------------------- Timing ---------------------
elapsedMillis roadTimer;
elapsedMillis soundTimer;

elapsedMillis uiTimer;
// -------------------- Game State -----------------
enum GameMode {
MODE_TITLE,
MODE_PLAYING,
MODE_CRASH
};
GameMode mode = MODE_TITLE;
// -------------------- Road Model -----------------
int roadCenter[SCREEN_HEIGHT];
// -------------------- Player ---------------------
float playerX = SCREEN_WIDTH / 2.0f;
constexpr int CAR_Y = SCREEN_HEIGHT - 10;
constexpr int CAR_W = 8;
constexpr int CAR_H = 6;
// -------------------- Joystick -------------------
int joystickCenter = 512;
constexpr int JOY_DEADBAND = 40;
// -------------------- Shifting / Speed ----------
int gear = 1;
constexpr int MAX_GEAR = 4;
float rpm = 1200.0f;
const float gearMinRPM[MAX_GEAR + 1] = {0, 1000, 1500, 1700, 1900};
const float gearMaxRPM[MAX_GEAR + 1] = {0, 4200, 5200, 6200, 7200};
const float gearAccel[MAX_GEAR + 1] = {0, 11.0, 9.5, 8.0, 6.5};
const float gearDrag[MAX_GEAR + 1] = {0, 1.5, 1.3, 1.2, 1.0};
const float turnAuthority[MAX_GEAR + 1] = {0, 2.0, 2.4, 2.8, 3.2};
int roadScrollMs = 80;
// -------------------- Button Debounce -----------
bool lastButtonReading = HIGH;
bool stableButtonState = HIGH;

unsigned long lastDebounceTime = 0;
constexpr unsigned long DEBOUNCE_MS = 30;
// -------------------- Sound ----------------------
bool shiftBeepActive = false;
unsigned long shiftBeepEnd = 0;
unsigned long crashStartTime = 0;
// 
============================================================
// Utility Functions
// 
============================================================

int clampInt(int x, int lo, int hi) {
if (x < lo) return lo;
if (x > hi) return hi;
return x;
}
float clampFloat(float x, float lo, float hi) {
if (x < lo) return lo;
if (x > hi) return hi;
return x;
}
float mapFloat(float x, float inMin, float inMax, float outMin, float
outMax) {
if (fabs(inMax - inMin) < 0.0001f) return outMin;
return (x - inMin) * (outMax - outMin) / (inMax - inMin) + outMin;
}
// 
============================================================
// Input
// 
============================================================
void calibrateJoystickCenter() {
long sum = 0;
const int samples = 40;
for (int i = 0; i < samples; i++) {
sum += analogRead(JOYSTICK_PIN);
delay(5);

}
joystickCenter = sum / samples;
}
bool buttonPressedEvent() {
bool reading = digitalRead(BUTTON_PIN);
if (reading != lastButtonReading) {
lastDebounceTime = millis();
}
if ((millis() - lastDebounceTime) > DEBOUNCE_MS) {
if (reading != stableButtonState) {
stableButtonState = reading;
if (stableButtonState == LOW) {
lastButtonReading = reading;
return true;
}
}
}
lastButtonReading = reading;
return false;
}
float readSteeringInput() {
int raw = analogRead(JOYSTICK_PIN);
int delta = raw - joystickCenter;
if (abs(delta) < JOY_DEADBAND) return 0.0f;
float steer = (float)delta / 512.0f;
steer = clampFloat(steer, -1.0f, 1.0f);
return steer;
}
// 
============================================================
// Road Generation
// 
============================================================

void initRoad() {
int startCenter = SCREEN_WIDTH / 2;
for (int y = 0; y < SCREEN_HEIGHT; y++) {
roadCenter[y] = startCenter;
}
}
void scrollRoad() {
for (int y = SCREEN_HEIGHT - 1; y > 0; y--) {
roadCenter[y] = roadCenter[y - 1];
}
static int currentCenter = SCREEN_WIDTH / 2;
int wander = random(-3, 4);
currentCenter += wander;
currentCenter = clampInt(currentCenter, 30, SCREEN_WIDTH - 30);
roadCenter[0] = currentCenter;
}
int currentRoadWidth() {
int width = 44 - (gear - 1) * 4;
return clampInt(width, 28, 44);
}

Project 1: Complete Code

Complete Code

bool playerOffRoad() {
int roadW = currentRoadWidth();
int center = roadCenter[CAR_Y + CAR_H / 2];
int leftEdge = center - roadW / 2;
int rightEdge = center + roadW / 2;
int carLeft = (int)(playerX - CAR_W / 2);
int carRight = (int)(playerX + CAR_W / 2);
return (carLeft < leftEdge || carRight > rightEdge);
}
// 
============================================================
// Sound / Servo
// 
============================================================

void speakerSelfTest() {
// Useful for immediate hardware verification
tone(SPEAKER_PIN, 1000);
delay(250);
tone(SPEAKER_PIN, 1600);
delay(250);
tone(SPEAKER_PIN, 2200);
delay(250);
noTone(SPEAKER_PIN);
delay(150);
}
void triggerShiftBeep() {
shiftBeepActive = true;
shiftBeepEnd = millis() + 100;
}
void updateSound() {
if (mode == MODE_TITLE) {
noTone(SPEAKER_PIN);
return;
}
if (mode == MODE_CRASH) {
static int crashFreq = 2200;
static unsigned long lastCrashStep = 0;
if (millis() - lastCrashStep > 40) {
lastCrashStep = millis();
crashFreq -= 180;
if (crashFreq < 700) crashFreq = 2200;
tone(SPEAKER_PIN, crashFreq);
}
return;
}
if (shiftBeepActive) {
if (millis() < shiftBeepEnd) {
tone(SPEAKER_PIN, 2800);

return;
} else {
shiftBeepActive = false;
}
}
// Higher range for tiny speaker / transistor setup
int freq = (int)mapFloat(rpm, 1000.0f, 7200.0f, 900.0f, 2600.0f);
freq = clampInt(freq, 900, 2600);
tone(SPEAKER_PIN, freq);
}
void updateServo() {
int angle = (int)mapFloat(rpm, 1000.0f, 7200.0f, 15.0f, 165.0f);
angle = clampInt(angle, 15, 165);
rpmServo.write(angle);
}
// 
============================================================
// Game Logic
// 
============================================================
void resetGame() {
gear = 1;
rpm = 1200.0f;
playerX = SCREEN_WIDTH / 2.0f;
roadScrollMs = 80;
initRoad();
shiftBeepActive = false;
mode = MODE_PLAYING;
}

void startCrash() {
mode = MODE_CRASH;
crashStartTime = millis();
}
void tryShiftUp() {
if (gear >= MAX_GEAR) return;

if (rpm > gearMaxRPM[gear] * 0.72f) {
gear++;
rpm *= 0.68f;
if (rpm < gearMinRPM[gear]) rpm = gearMinRPM[gear];
triggerShiftBeep();
} else {
rpm -= 250.0f;
if (rpm < 1000.0f) rpm = 1000.0f;
triggerShiftBeep();
}
}
void updatePhysics(float dt) {
rpm += gearAccel[gear] * dt * 100.0f;
rpm -= gearDrag[gear] * dt * 20.0f;
if (rpm > gearMaxRPM[gear]) {
rpm -= 10.0f * dt * 100.0f;
}
rpm = clampFloat(rpm, 1000.0f, 7200.0f);
float steer = readSteeringInput();
float speedFactor = mapFloat(rpm, 1000.0f, 7200.0f, 0.5f, 1.35f);
playerX += steer * turnAuthority[gear] * speedFactor;
playerX = clampFloat(playerX, 4.0f, SCREEN_WIDTH - 4.0f);
roadScrollMs = (int)mapFloat(rpm, 1000.0f, 7200.0f, 95.0f, 28.0f);
roadScrollMs -= (gear - 1) * 4;
roadScrollMs = clampInt(roadScrollMs, 20, 95);
}
// 
============================================================
// Drawing
// 
============================================================
void drawCar(int x, int y) {
display.fillRect(x - 3, y - 2, 6, 4, SSD1306_WHITE);
display.drawPixel(x - 2, y - 3, SSD1306_WHITE);

display.drawPixel(x + 2, y - 3, SSD1306_WHITE);
display.drawPixel(x - 2, y + 3, SSD1306_WHITE);
display.drawPixel(x + 2, y + 3, SSD1306_WHITE);
}
void drawRoad() {
int roadW = currentRoadWidth();
for (int y = 0; y < SCREEN_HEIGHT; y++) {
int center = roadCenter[y];
int leftEdge = center - roadW / 2;
int rightEdge = center + roadW / 2;
display.drawPixel(leftEdge, y, SSD1306_WHITE);
display.drawPixel(rightEdge, y, SSD1306_WHITE);
if ((y / 4) % 2 == 0) {
display.drawPixel(center, y, SSD1306_WHITE);
}
}
}
void drawHUD() {
display.setTextSize(1);
display.setTextColor(SSD1306_WHITE);
display.setCursor(2, 2);
display.print("G:");
display.print(gear);
display.setCursor(38, 2);
display.print("RPM:");
display.print((int)rpm);
}
void drawTitleScreen() {
display.clearDisplay();
display.setTextSize(1);
display.setTextColor(SSD1306_WHITE);

display.setCursor(22, 14);
display.println("TEENSY RACER V1");
display.setCursor(14, 30);
display.println("Joystick = steer");
display.setCursor(14, 40);
display.println("Button = shift/start");
display.setCursor(18, 54);
display.println("Press to begin");
display.display();
}

Project 1: Complete Code

Complete Code

void drawCrashScreen() {
display.clearDisplay();
drawRoad();
drawCar((int)playerX, CAR_Y);
display.fillRect(18, 22, 92, 18, SSD1306_BLACK);
display.drawRect(18, 22, 92, 18, SSD1306_WHITE);
display.setTextSize(1);
display.setTextColor(SSD1306_WHITE);
display.setCursor(42, 27);
display.print("CRASH!");
display.setCursor(22, 44);
display.print("Press to retry");
display.display();
}
void drawGame() {
display.clearDisplay();
drawRoad();
drawCar((int)playerX, CAR_Y);
drawHUD();
display.display();

}
// 
============================================================
// Setup / Loop
// 
============================================================
void setup() {
pinMode(BUTTON_PIN, INPUT_PULLUP);
pinMode(SPEAKER_PIN, OUTPUT);
digitalWrite(SPEAKER_PIN, LOW);
analogReadResolution(10);
Wire.begin();
Wire.setClock(400000);
if (!display.begin(SSD1306_SWITCHCAPVCC, 0x3C)) {
while (true) {
delay(1000);
}
}
display.clearDisplay();
display.display();
rpmServo.attach(SERVO_PIN);
rpmServo.write(15);
randomSeed(analogRead(JOYSTICK_PIN) + micros());
calibrateJoystickCenter();
initRoad();
// Speaker hardware verification at boot
speakerSelfTest();
mode = MODE_TITLE;
}
void loop() {

if (buttonPressedEvent()) {
if (mode == MODE_TITLE) {
resetGame();
} else if (mode == MODE_PLAYING) {
tryShiftUp();
} else if (mode == MODE_CRASH) {
resetGame();
}
}
if (mode == MODE_PLAYING) {
static unsigned long lastPhysicsMs = millis();
unsigned long now = millis();
float dt = (now - lastPhysicsMs) / 1000.0f;
lastPhysicsMs = now;
dt = clampFloat(dt, 0.0f, 0.05f);
updatePhysics(dt);
if (roadTimer >= (unsigned long)roadScrollMs) {
roadTimer = 0;
scrollRoad();
if (playerOffRoad()) {
startCrash();
}
}
updateServo();
if (uiTimer >= 16) {
uiTimer = 0;
drawGame();
}
}
else if (mode == MODE_TITLE) {
drawTitleScreen();
rpm = 1000.0f;
updateServo();

}
else if (mode == MODE_CRASH) {
if (millis() - crashStartTime > 1200) {
drawCrashScreen();
noTone(SPEAKER_PIN);
rpmServo.write(20);
} else {
drawCrashScreen();
}
}
if (soundTimer >= 20) {
soundTimer = 0;
updateSound();
}
}

Project 2: Complete Code

Complete Code

// Nightlight System
// Photoresistor + Ultrasonic + RGB LED + Gentle Melody
// Teensy 4.1
int photoPin = A0; int trigPin = 7; int echoPin = 9; int redPin = 24; int greenPin = 25; 
int bluePin = 26; int buzzerPin = 6; int lightValue = 0; int darkLevel = 500; int 
closeDistance = 50; float distance = 0;
bool alreadyTriggered = false;
void setup() {
  pinMode(trigPin, OUTPUT); pinMode(echoPin, INPUT); pinMode(redPin, OUTPUT);
  pinMode(greenPin, OUTPUT); pinMode(bluePin, OUTPUT); pinMode(buzzerPin, 
OUTPUT);
  // Start RGB LED off, setLED(false); Serial.begin(9600); analogReadResolution(10);
}
void loop() { lightValue = analogRead(photoPin); distance = getDistance(); 
Serial.print("Light: "); Serial.print(lightValue); Serial.print("   Distance: "); 
Serial.println(distance);
if (
    lightValue < darkLevel &&
    distance > 0 &&
    distance < closeDistance) {
// Same behavior as old single-color LED: turn the light ON. All three RGB channels 
create white.
setLED(true);
    if (!alreadyTriggered) { playLullaby(); alreadyTriggered = true;
    }
  } else {
    setLED(false);
    noTone(buzzerPin);
    alreadyTriggered = false;
  }
  delay(100);
}
void setLED(bool state) {
  if (state) {
    digitalWrite(redPin, HIGH);
    digitalWrite(greenPin, HIGH);
    digitalWrite(bluePin, HIGH);
  } else {
    digitalWrite(redPin, LOW);
    digitalWrite(greenPin, LOW);
    digitalWrite(bluePin, LOW);
  }
}
float getDistance() {

  long time;
  digitalWrite(trigPin, LOW);
  delayMicroseconds(2);
  digitalWrite(trigPin, HIGH);
  delayMicroseconds(10);
  digitalWrite(trigPin, LOW);
  time = pulseIn(
    echoPin,
    HIGH,
    30000
  );
  return time * 0.034 / 2;
}
void playLullaby() {
  tone(buzzerPin, 523, 250);
  delay(300);
  tone(buzzerPin, 587, 250);
  delay(300);
  tone(buzzerPin, 659, 300);
  delay(350);
  tone(buzzerPin, 698, 300);
  delay(350);
  tone(buzzerPin, 659, 300);
  delay(350);
  tone(buzzerPin, 587, 250);
  delay(300);
  tone(buzzerPin, 523, 350);
  delay(400);
  tone(buzzerPin, 587, 250);
  delay(300);
  tone(buzzerPin, 523, 500);
  delay(550);
  noTone(buzzerPin);
}

EMBEDDED SYSTEMS LEARNING KIT

R E S O U R C E

AI Game Designer Prompt

Copy and paste this prompt into ChatGPT, Claude, or your preferred AI assistant to start designing your own game.

SYSTEM PROMPT TEMPLATE COPY PROMPT

You are my AI game designer and embedded-systems coding partner.

I am using the North Star Scientific Embedded Systems Learning Kit. Read the attached learning guide first, especially Mini Project 1 and its complete code, so you 

understand the available hardware, wiring, Teensy 4.1, OLED display, joystick, push button, servo, buzzer/speaker, and existing libraries.

Your goal is to help me invent and build my own fun, creative, interactive game using this hardware.

Step 1: Design the game with me

Lead me through 5–10 short questions, one at a time. Ask about things like:

• What type of game sounds fun to me

• Theme or setting

• What the player should control

• What the goal should be

• Enemies, obstacles, challenges, or puzzles

• How scoring or winning should work

• How the joystick and button could be used

• How the servo and sound could make the game more interactive

• Difficulty, levels, power-ups, surprises, or special mechanics

Be very creative and open-ended. Do not limit ideas to racing games. If I don't know an answer, always give me 3–5 specific, interesting ideas I can choose from. 

Examples could include spaceship battles, dungeon games, fishing games, survival games, reaction games, maze games, rhythm games, monster battles, sports games, 

treasure hunts, stealth games, or completely original ideas.

Step 2: Turn the idea into a game

After you understand my idea, briefly show me:

• Game Name

• Concept

• Controls

• Goal

• Game Mechanics

• How each piece of hardware is used

Let me approve or change the idea before coding.

Step 3: Vibe-code it

Once I approve it, generate the complete Arduino/Teensy code needed to play the game. Use the attached guide's existing hardware setup, pins, libraries, display 

configuration, and wiring wherever possible. Do not require extra components unless you clearly label them as optional.

Make the code:

• Complete and copy/paste ready

• Beginner-friendly

• Commented clearly

• Fun and polished rather than just a technical demo

• Designed to run smoothly on the Teensy 4.1

• Compatible with the kit's OLED, joystick, button, servo, and buzzer/speaker

Then give me very short instructions for:

• Copying the code into Arduino IDE

• Uploading it

• Playing the game

• Fixing the most likely problems

After it works, suggest 3 cool upgrades I could ask you to add next.

Start by reading the guide and then ask me Question 1.

